

Software Design Notes

1. Revision History

Rev.	Revision Notes	DCN	Date
1	Initial Release	N/A	24SEP2021
2			

2. Introduction

2.1. Purpose

This document presents software design notes for a VBA subroutine that logs a form and saves the form as a PDF.

This software is intended to automatically log and file the Observation Form, 100-001.

Minimal risk is associated with this software.

2.2. Scope

This document applies to the VBA subroutine(s) listed in Table 1 – VBA Form Submission Subroutine.

Table 1 – VBA Form Submission Subroutine

Subroutine	Description
submitButton_Click()	<ul style="list-style-type: none">• Adds a row to an Excel file located in OBSERVATION_LOG_FILEPATHNAME.• Unprotects/Protects the Excel file with OBSERVATION_LOG_PASSWORD.• Saves a PDF in PDF_FILEPATH.• The subroutine must be located in the ThisDocument.cls file.

2.3. References

2.3.1. File Locations

Published Files

A controlled Observation Form is available in the GitHub repository.

- <https://github.com/nolan210/BexarVBA.git>

VBA Source Code

A text file of the ThisDocument.cls source code is available in the GitHub repository.

- <https://github.com/nolan210/BexarVBA.git>

Engineering Files

A copy of the Word Macro Enabled Template, Excel file and VBA Source Code is in the GitHub repository.

- <https://github.com/nolan210/BexarVBA.git>

3. Software Specification Requirements

SSR1 – Usability

No text (title)

SSR1.1

Embedded in a Word Macro Enabled Template.

SSR1.2

Accesses and saves files on a local file server.

SSR1.3

Does not require the document user to locate files or navigate folders.

SSR1.4

A single “Submit” button accomplishes all logging and file saving functionality.

SSR2 – Functionality

No text (title)

SSR2.1

Log date and time at button click.

SSR2.2

Log username of account logged into computer at button click.

SSR2.3

Log computer ID at button click.

SSR2.4

Log a primary key at button click, which is equal to the row index of the form’s Excel entry.

SSR2.5

Log the Observation Form’s text form fields in the Excel file.

SSR2.6

Save the Observation Form as a PDF document.

SSR2.7


Unprotect the sheet immediately prior to logging.

SSR2.8

Protect the sheet immediately after logging.

4. Flowchart

submitButton_Click()



5. Source Code

Table 2 – VBA Source Code

```
VERSION 1.0 CLASS
BEGIN
    MultiUse = -1    'True
END
Attribute VB_Name = "ThisDocument"
Attribute VB_GlobalNameSpace = False
Attribute VB_Creatable = False
Attribute VB_PredeclaredId = True
Attribute VB_Exposed = True
'Used to move documents
Private Const PDF_FILEPATH As String = "C:\Repos\Software\Bexar VBA\BexarVBA\100\001 Observation
Form\Observation Log\PDFs\"
Private Const OBSERVATION_LOG_FILEPATHNAME As String = "C:\Repos\Software\Bexar
VBA\BexarVBA\100\001 Observation Form\Observation Log\Observation Log.xlsx"
Private Const OBSERVATION_LOG_PASSWORD As String = "Z45KUWkf37Drdy*2"

'Used to access Content Controls
Private Const oTitle = "observationTitle"
Private Const oType = "observationType"
Private Const oObservation = "observation"
Private Const oCaption1 = "caption1"
Private Const oCaption2 = "caption2"

'Used to access data array
Enum formInputEnums
    iUser
    iComputer
    iDatetime
    iTitle
    iType
    iObservation
    iCaption1
    iCaption2
    iPDF
End Enum

Private Sub submitButton_Click()
'Nolan Manteufel
'10JUL2020

'Logs form in OBSERVATION_LOG_FILEPATHNAME excel file.
'Protects the OBSERVATION_LOG_FILEPATHNAME excel file with OBSERVATION_LOG_PASSWORD.
'Saves pdf of form in PDF_FILEPATH folder.

'Variables
Dim formDoc As Document
Set formDoc = ActiveDocument
Dim cc As ContentControls

Dim excelApp As Excel.Application
Dim logWB As Excel.Workbook
Dim logSheet As Excel.Worksheet

'User
Dim formUser As String
formUser = Environ$("username")
formUser = LCase(formUser)
```

```
'Computer
Dim formComputer As String
formComputer = Environ$("computername")
formComputer = LCase(formComputer)

'Datetime
Dim dateTime As String
Dim timeOnly As String
timeOnly = Format(Time, "hhmmss")
dateTime = Format(Date, "yyyymmdd")
dateTime = dateTime & timeOnly

'Inputs array
Dim formInputs(8) As String

'Populate metadata
formInputs(iUser) = formUser
formInputs(iComputer) = formComputer
formInputs(iDatetime) = dateTime

'Populate data
Set cc = formDoc.SelectContentControlsByTag(oTitle)
formInputs(iTitle) = cc(1).Range.Text

Set cc = formDoc.SelectContentControlsByTag(oType)
formInputs(iType) = cc(1).Range.Text

Set cc = formDoc.SelectContentControlsByTag(oObservation)
formInputs(iObservation) = cc(1).Range.Text

Set cc = formDoc.SelectContentControlsByTag(oCaption1)
formInputs(iCaption1) = cc(1).Range.Text

Set cc = formDoc.SelectContentControlsByTag(oCaption2)
formInputs(iCaption2) = cc(1).Range.Text

formInputs(iPDF) = PDF_FILEPATH & dateTime & "_" & formUser & ".pdf"

'Launch excel
Set excelApp = New Excel.Application

'The workbook
Set logWB = excelApp.Workbooks.Open(FileName:=OBSERVATION_LOG_FILEPATHNAME)

'The worksheet
Set logSheet = logWB.Worksheets("Observations")

'Unprotect the log
logSheet.Unprotect (OBSERVATION_LOG_PASSWORD)

'Log the entry
Dim observationPKey As Long
observationPKey = logSheet.UsedRange.Rows.Count + 1
logSheet.Cells(observationPKey, 1).Value = observationPKey
logSheet.Cells(observationPKey, 2).Value = formInputs(iDatetime)
logSheet.Cells(observationPKey, 3).Value = formInputs(iUser)
logSheet.Cells(observationPKey, 4).Value = formInputs(iComputer)
logSheet.Cells(observationPKey, 5).Value = formInputs(iType)
logSheet.Cells(observationPKey, 6).Value = formInputs(iTitle)
logSheet.Cells(observationPKey, 7).Value = formInputs(iObservation)
logSheet.Cells(observationPKey, 8).Value = formInputs(iCaption1)
logSheet.Cells(observationPKey, 9).Value = formInputs(iCaption2)
```

```
logSheet.Cells(observationPKey, 10).Value = formInputs(iPDF)

'Protect the log
logSheet.Protect (OBSERVATION_LOG_PASSWORD)

'Close log
logWB.Close savechanges:=True

'Release object references
Set cc = Nothing
Set logSheet = Nothing
Set logWB = Nothing
Set excelApp = Nothing

'Save as PDF
formDoc.ExportAsFixedFormat _
    OutputFileName:=formInputs(iPDF), _
    ExportFormat:=wdExportFormatPDF, _
    OpenAfterExport:=False

'Confirmation message
MsgBox ("Awesome, " & formUser & ". Your observation has been submitted successfully.")

End Sub
```